pragma solidity ^0.4.23;

contract Owned {

    event OwnerChanged(address indexed from, address indexed to);

    address public owner;

    constructor() public {
        owner = msg.sender;
    }

    modifier onlyOwner {
        require(msg.sender == owner);
        _;
    }

    function _transferOwnership(address _from, address _to) internal {
        owner = _to;
        emit OwnerChanged(_from, _to);
    }

    function transferOwnership(address newOwner) onlyOwner public {
        _transferOwnership(owner, newOwner);
    }
}

contract Whitelisted is Owned {

    event WhitelistModified(address indexed who, bool inWhitelist);

    mapping(address => bool) public whitelist;

    constructor() public {
        whitelist[msg.sender] = true;
    }

    function addToWhitelist(address who) public onlyOwner {
        whitelist[who] = true;
        emit WhitelistModified(who, true);
    }
    
    function removeFromWhitelist(address who) public onlyOwner {
        whitelist[who] = false;
        emit WhitelistModified(who, false);
    }

    modifier whitelisted {
        require(whitelist[msg.sender] == true);
        _;
    }

}

contract TokenERC20 {
    // Public variables of the token
    string public name;
    string public symbol;
    uint8 public decimals = 18;
    // 18 decimals is the strongly suggested default, avoid changing it
    uint256 public totalSupply;

    // This creates an array with all balances
    mapping (address => uint256) public balanceOf;
    mapping (address => mapping (address => uint256)) public allowance;

    // This generates a public event on the blockchain that will notify clients
    event Transfer(address indexed from, address indexed to, uint256 value);
    
    // This generates a public event on the blockchain that will notify clients
    event Approval(address indexed _owner, address indexed _spender, uint256 _value);

    // This notifies clients about the amount burnt
    event Burn(address indexed from, uint256 value);

    /**
     * Constructor function
     *
     * Initializes contract with initial supply tokens to the creator of the contract
     */
    constructor(
        uint256 initialSupply,
        string tokenName,
        string tokenSymbol
    ) public {
        totalSupply = initialSupply * 10 ** uint256(decimals);  // Update total supply with the decimal amount
        balanceOf[msg.sender] = totalSupply;                // Give the creator all initial tokens
        name = tokenName;                                   // Set the name for display purposes
        symbol = tokenSymbol;                               // Set the symbol for display purposes
    }

    /**
     * Internal transfer, only can be called by this contract
     */
    function _transfer(address _from, address _to, uint _value) internal {
        // Prevent transfer to 0x0 address. Use burn() instead
        require(_to != 0x0);
        // Check if the sender has enough
        require(balanceOf[_from] >= _value);
        // Check for overflows
        require(balanceOf[_to] + _value >= balanceOf[_to]);
        // Save this for an assertion in the future
        uint previousBalances = balanceOf[_from] + balanceOf[_to];
        // Subtract from the sender
        balanceOf[_from] -= _value;
        // Add the same to the recipient
        balanceOf[_to] += _value;
        emit Transfer(_from, _to, _value);
        // Asserts are used to use static analysis to find bugs in your code. They should never fail
        assert(balanceOf[_from] + balanceOf[_to] == previousBalances);
    }

    /**
     * Transfer tokens
     *
     * Send `_value` tokens to `_to` from your account
     *
     * @param _to The address of the recipient
     * @param _value the amount to send
     */
    function transfer(address _to, uint256 _value) public returns (bool success) {
        _transfer(msg.sender, _to, _value);
        return true;
    }

    /**
     * Transfer tokens from other address
     *
     * Send `_value` tokens to `_to` on behalf of `_from`
     *
     * @param _from The address of the sender
     * @param _to The address of the recipient
     * @param _value the amount to send
     */
    function transferFrom(address _from, address _to, uint256 _value) public returns (bool success) {
        require(_value <= allowance[_from][msg.sender]);     // Check allowance
        allowance[_from][msg.sender] -= _value;
        _transfer(_from, _to, _value);
        return true;
    }

    /**
     * Set allowance for other address
     *
     * Allows `_spender` to spend no more than `_value` tokens on your behalf
     *
     * @param _spender The address authorized to spend
     * @param _value the max amount they can spend
     */
    function approve(address _spender, uint256 _value) public
        returns (bool success) {
        allowance[msg.sender][_spender] = _value;
        emit Approval(msg.sender, _spender, _value);
        return true;
    }

    /**
     * Destroy tokens
     *
     * Remove `_value` tokens from the system irreversibly
     *
     * @param _value the amount of money to burn
     */
    function burn(uint256 _value) public returns (bool success) {
        require(balanceOf[msg.sender] >= _value);   // Check if the sender has enough
        balanceOf[msg.sender] -= _value;            // Subtract from the sender
        totalSupply -= _value;                      // Updates totalSupply
        emit Burn(msg.sender, _value);
        return true;
    }

    /**
     * Destroy tokens from other account
     *
     * Remove `_value` tokens from the system irreversibly on behalf of `_from`.
     *
     * @param _from the address of the sender
     * @param _value the amount of money to burn
     */
    function burnFrom(address _from, uint256 _value) public returns (bool success) {
        require(balanceOf[_from] >= _value);                // Check if the targeted balance is enough
        require(_value <= allowance[_from][msg.sender]);    // Check allowance
        balanceOf[_from] -= _value;                         // Subtract from the targeted balance
        allowance[_from][msg.sender] -= _value;             // Subtract from the sender's allowance
        totalSupply -= _value;                              // Update totalSupply
        emit Burn(_from, _value);
        return true;
    }

}

contract Ellobitz is TokenERC20, Owned, Whitelisted {

    uint256 public mineCount;
    uint256 public minMineSize;
    uint256 public maxMineSize;
    uint256 public chipSize;
    uint256 public firstChipBonus;
    uint public chipSpeed;

    uint256 internal activeMine;
    uint256 internal mineSize;
    bool internal firstChip;
    
    mapping(address => uint) public lastChipTime;

    event MineFound(address indexed chipper, uint256 activeMine);
    event MineChipped(address indexed chipper, uint256 indexed activeMine, uint256 amount);
    event MineExausted(address indexed chipper, uint256 activeMine);

    modifier validMineParameters (
        uint256 _mineCount,
        uint256 _minMineSize,
        uint256 _maxMineSize,
        uint256 _chipSize,
        uint256 _firstChipBonus,
        uint _chipSpeed
    ) {
        require(_minMineSize <= _maxMineSize, "Smallest mine size smaller than largest mine size");
        require(_chipSize + _firstChipBonus <= _minMineSize, "First chip would exhaust mine");
        _;
    }

    constructor(
        string tokenName,
        string tokenSymbol,
        uint256 _mineCount,
        uint256 _minMineSize,
        uint256 _maxMineSize,
        uint256 _chipSize,
        uint256 _firstChipBonus,
        uint _chipSpeed
    ) TokenERC20(0, tokenName, tokenSymbol) validMineParameters(
        _mineCount,
        _minMineSize,
        _maxMineSize,
        _chipSize,
        _firstChipBonus,
        _chipSpeed
    ) public {
        
        // variable setting
        mineCount = _mineCount;
        minMineSize = _minMineSize;
        maxMineSize = _maxMineSize;
        chipSize = _chipSize;
        firstChipBonus = _firstChipBonus;
        chipSpeed = _chipSpeed;

        // other variable initialization
        activeMine = 0;
        mineSize = minMineSize;
        firstChip = true;
    }

    function _resetMine() internal {
        activeMine = random() % mineCount;
        mineSize = random() % (maxMineSize - minMineSize + 1) + minMineSize;
        firstChip = true;
    }

    function chip(uint256 mineNumber) public whitelisted {
        
        require(mineNumber == activeMine, "Chipped wrong mine");
        require(now >= lastChipTime[msg.sender] + chipSpeed, "Chipped too fast");
        
        uint256 thisChipNoCap = firstChip ? firstChipBonus + chipSize : chipSize;
        uint256 thisChip = thisChipNoCap > mineSize ? mineSize : thisChipNoCap;

        if (firstChip) {
            emit MineFound(msg.sender, activeMine);
        }

        mineSize -= thisChip;
        mintToken(msg.sender, thisChip);
        lastChipTime[msg.sender] = now;
        firstChip = false;
        emit MineChipped(msg.sender, activeMine, thisChip);

        if (mineSize <= 0) {
            emit MineExausted(msg.sender, activeMine);
            _resetMine();
        }
    }

    function setParameters(
        uint256 _mineCount,
        uint256 _minMineSize,
        uint256 _maxMineSize,
        uint256 _chipSize,
        uint256 _firstChipBonus,
        uint _chipSpeed
    ) onlyOwner validMineParameters(
        _mineCount,
        _minMineSize,
        _maxMineSize,
        _chipSize,
        _firstChipBonus,
        _chipSpeed
    ) public {
        mineCount = _mineCount;
        minMineSize = _minMineSize;
        maxMineSize = _maxMineSize;
        chipSize = _chipSize;
        firstChipBonus = _firstChipBonus;
        chipSpeed = _chipSpeed;
    }

    function mintToken(address target, uint256 mintedAmount) internal {
        balanceOf[target] += mintedAmount;
        totalSupply += mintedAmount;
        emit Transfer(this, target, mintedAmount);
    }

    // adapted from https://medium.com/@promentol/lottery-smart-contract-can-we-generate-random-numbers-in-solidity-4f586a152b27
    function random() internal view returns (uint256) {
        return uint256(keccak256(
            abi.encodePacked(block.timestamp, block.difficulty)
        ));
    }

}